creation_sourcecode: // File: @openzeppelin/contracts/utils/Context.sol
  
  
  // OpenZeppelin Contracts (last updated v5.0.1) (utils/Context.sol)
  
  pragma solidity ^0.8.20;
  
  /**
   * @dev Provides information about the current execution context, including the
   * sender of the transaction and its data. While these are generally available
   * via msg.sender and msg.data, they should not be accessed in such a direct
   * manner, since when dealing with meta-transactions the account sending and
   * paying for execution may not be the actual sender (as far as an application
   * is concerned).
   *
   * This contract is only required for intermediate, library-like contracts.
   */
  abstract contract Context {
      function _msgSender() internal view virtual returns (address) {
          return msg.sender;
      }
  
      function _msgData() internal view virtual returns (bytes calldata) {
          return msg.data;
      }
  
      function _contextSuffixLength() internal view virtual returns (uint256) {
          return 0;
      }
  }
  
  // File: @openzeppelin/contracts/access/Ownable.sol
  
  
  // OpenZeppelin Contracts (last updated v5.0.0) (access/Ownable.sol)
  
  pragma solidity ^0.8.20;
  
  
  /**
   * @dev Contract module which provides a basic access control mechanism, where
   * there is an account (an owner) that can be granted exclusive access to
   * specific functions.
   *
   * The initial owner is set to the address provided by the deployer. This can
   * later be changed with {transferOwnership}.
   *
   * This module is used through inheritance. It will make available the modifier
   * `onlyOwner`, which can be applied to your functions to restrict their use to
   * the owner.
   */
  abstract contract Ownable is Context {
      address private _owner;
  
      /**
       * @dev The caller account is not authorized to perform an operation.
       */
      error OwnableUnauthorizedAccount(address account);
  
      /**
       * @dev The owner is not a valid owner account. (eg. `address(0)`)
       */
      error OwnableInvalidOwner(address owner);
  
      event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);
  
      /**
       * @dev Initializes the contract setting the address provided by the deployer as the initial owner.
       */
      constructor(address initialOwner) {
          if (initialOwner == address(0)) {
              revert OwnableInvalidOwner(address(0));
          }
          _transferOwnership(initialOwner);
      }
  
      /**
       * @dev Throws if called by any account other than the owner.
       */
      modifier onlyOwner() {
          _checkOwner();
          _;
      }
  
      /**
       * @dev Returns the address of the current owner.
       */
      function owner() public view virtual returns (address) {
          return _owner;
      }
  
      /**
       * @dev Throws if the sender is not the owner.
       */
      function _checkOwner() internal view virtual {
          if (owner() != _msgSender()) {
              revert OwnableUnauthorizedAccount(_msgSender());
          }
      }
  
      /**
       * @dev Leaves the contract without owner. It will not be possible to call
       * `onlyOwner` functions. Can only be called by the current owner.
       *
       * NOTE: Renouncing ownership will leave the contract without an owner,
       * thereby disabling any functionality that is only available to the owner.
       */
      function renounceOwnership() public virtual onlyOwner {
          _transferOwnership(address(0));
      }
  
      /**
       * @dev Transfers ownership of the contract to a new account (`newOwner`).
       * Can only be called by the current owner.
       */
      function transferOwnership(address newOwner) public virtual onlyOwner {
          if (newOwner == address(0)) {
              revert OwnableInvalidOwner(address(0));
          }
          _transferOwnership(newOwner);
      }
  
      /**
       * @dev Transfers ownership of the contract to a new account (`newOwner`).
       * Internal function without access restriction.
       */
      function _transferOwnership(address newOwner) internal virtual {
          address oldOwner = _owner;
          _owner = newOwner;
          emit OwnershipTransferred(oldOwner, newOwner);
      }
  }
  
  // File: @openzeppelin/contracts/security/ReentrancyGuard.sol
  
  
  // OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)
  
  pragma solidity ^0.8.0;
  
  /**
   * @dev Contract module that helps prevent reentrant calls to a function.
   *
   * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
   * available, which can be applied to functions to make sure there are no nested
   * (reentrant) calls to them.
   *
   * Note that because there is a single `nonReentrant` guard, functions marked as
   * `nonReentrant` may not call one another. This can be worked around by making
   * those functions `private`, and then adding `external` `nonReentrant` entry
   * points to them.
   *
   * TIP: If you would like to learn more about reentrancy and alternative ways
   * to protect against it, check out our blog post
   * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
   */
  abstract contract ReentrancyGuard {
      // Booleans are more expensive than uint256 or any type that takes up a full
      // word because each write operation emits an extra SLOAD to first read the
      // slot's contents, replace the bits taken up by the boolean, and then write
      // back. This is the compiler's defense against contract upgrades and
      // pointer aliasing, and it cannot be disabled.
  
      // The values being non-zero value makes deployment a bit more expensive,
      // but in exchange the refund on every call to nonReentrant will be lower in
      // amount. Since refunds are capped to a percentage of the total
      // transaction's gas, it is best to keep them low in cases like this one, to
      // increase the likelihood of the full refund coming into effect.
      uint256 private constant _NOT_ENTERED = 1;
      uint256 private constant _ENTERED = 2;
  
      uint256 private _status;
  
      constructor() {
          _status = _NOT_ENTERED;
      }
  
      /**
       * @dev Prevents a contract from calling itself, directly or indirectly.
       * Calling a `nonReentrant` function from another `nonReentrant`
       * function is not supported. It is possible to prevent this from happening
       * by making the `nonReentrant` function external, and making it call a
       * `private` function that does the actual work.
       */
      modifier nonReentrant() {
          _nonReentrantBefore();
          _;
          _nonReentrantAfter();
      }
  
      function _nonReentrantBefore() private {
          // On the first call to nonReentrant, _status will be _NOT_ENTERED
          require(_status != _ENTERED, "ReentrancyGuard: reentrant call");
  
          // Any calls to nonReentrant after this point will fail
          _status = _ENTERED;
      }
  
      function _nonReentrantAfter() private {
          // By storing the original value once again, a refund is triggered (see
          // https://eips.ethereum.org/EIPS/eip-2200)
          _status = _NOT_ENTERED;
      }
  
      /**
       * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
       * `nonReentrant` function in the call stack.
       */
      function _reentrancyGuardEntered() internal view returns (bool) {
          return _status == _ENTERED;
      }
  }
  
  // File: contracts/LocEthV2.sol
  
  
  pragma solidity ^0.8.20;
  
  
  
  /**
   * @title TimedLockBox
   * @dev A smart contract for locking ETH with penalties over a specified period to prevent overtrading.
   */
  contract TimedLockBox is ReentrancyGuard, Ownable {
  
      /**
       * @dev Struct representing a lock for a specific wallet.
       */
      struct Lock {
          uint256 totalAmount;   // Total ETH locked by the wallet
          uint256 lockPeriod;    // Duration for which the ETH is locked
          uint256 unlockTime;    // Timestamp when the lock period ends
      }
  
      mapping(address => Lock) public locks;  // Mapping to store locks for each wallet
      uint256 public penaltyPercentage;       // Penalty percentage applied on early withdrawals
      uint256 public totalFees;               // Total fees accumulated from penalties
  
      // Events to log key contract activities
      event EthLocked(address indexed wallet, uint256 amount, uint256 lockPeriod);
      event EthUnlocked(address indexed wallet, uint256 amount, uint256 penaltyAmount);
      event FeesWithdrawn(address indexed owner, uint256 amount);
  
      /**
       * @dev Constructor for initializing the contract.
       * @param _penaltyPercentage The initial penalty percentage for early withdrawals.
       */
      constructor(uint _penaltyPercentage, address _owner) Ownable(_owner) {
          penaltyPercentage = _penaltyPercentage;
      }
  
      /**
       * @dev Lock ETH for a specified period.
       * @param lockPeriod Duration for which the ETH will be locked.
       */
      function lockEth(uint256 lockPeriod) external payable {
          require(msg.value > 0, "ETH amount must be greater than 0");
          require(lockPeriod > 0, "Lock period must be greater than 0");
  
          Lock storage userLock = locks[msg.sender];
  
          if (userLock.totalAmount > 0) {
              require(lockPeriod >= userLock.unlockTime - block.timestamp, "New lock period cannot be less than existing unlock time");
              userLock.totalAmount += msg.value;
              userLock.lockPeriod = lockPeriod;
              userLock.unlockTime = block.timestamp + lockPeriod;
          } else {
              locks[msg.sender] = Lock({
                  totalAmount: msg.value,
                  lockPeriod: lockPeriod,
                  unlockTime: block.timestamp + lockPeriod
              });
          }
  
          emit EthLocked(msg.sender, msg.value, lockPeriod);
      }
  
      /**
       * @dev Unlock ETH, applying penalties if the lock period has not ended.
       * @param forceUnlock Flag indicating whether to force the unlock without waiting for the lock period to end.
       */
      function unlockEth(bool forceUnlock) external nonReentrant {
          Lock storage userLock = locks[msg.sender];
          require(userLock.totalAmount > 0, "No locked ETH in the wallet");
  
          uint256 penaltyAmount = 0;
  
          if (!forceUnlock) {
              require(block.timestamp >= userLock.unlockTime, "Lock period not ended");
          }
  
          if (forceUnlock && block.timestamp < userLock.unlockTime) {
              penaltyAmount = (userLock.totalAmount * penaltyPercentage) / 100;
          }
  
          uint256 withdrawAmount = userLock.totalAmount - penaltyAmount;
  
          delete locks[msg.sender];
  
          payable(msg.sender).transfer(withdrawAmount);
  
          // Update total fees
          totalFees += penaltyAmount;
  
          emit EthUnlocked(msg.sender, withdrawAmount, penaltyAmount);
      }
  
      /**
       * @dev Withdraw accumulated fees, accessible only by the contract owner.
       */
      function withdrawFees() external onlyOwner {
          require(totalFees > 0, "No fees to withdraw");
  
          // Transfer fees to the owner
          payable(owner()).transfer(totalFees);
  
          // Reset total fees
          totalFees = 0;
  
          emit FeesWithdrawn(owner(), totalFees);
      }
  
      /**
       * @dev Get the unlock time and locked ETH amount for a specific wallet.
       * @param wallet The wallet address to query.
       * @return The unlock time and locked ETH amount.
       */
      function getUnlockTimeAndLockedEth(address wallet) external view returns (uint256, uint256) {
          Lock storage userLock = locks[wallet];
          return (userLock.unlockTime, userLock.totalAmount);
      }
  
  }